

# Machin's Series Expansion for $\pi$

Helmut Knaust, Department of Mathematical Sciences, UTEP, El Paso TX 79968  
hknaust@utep.edu  
8/8/09

## Gregory's Approximation

James Gregory (1638 - 1675) discovered the Taylor series expansion with center 0 for the arctan function:

**Series[ArcTan[x], {x, 0, 21}]**

$$x - \frac{x^3}{3} + \frac{x^5}{5} - \frac{x^7}{7} + \frac{x^9}{9} - \frac{x^{11}}{11} + \frac{x^{13}}{13} - \frac{x^{15}}{15} + \frac{x^{17}}{17} - \frac{x^{19}}{19} + \frac{x^{21}}{21} + O(x^{22})$$

One can use this series to approximate  $\pi$ . Note that  $\pi = 4 \arctan(1)$ .

**ArcTan[1]**

**4 ArcTan[1.]**

$$\frac{\pi}{4}$$

3.14159

Convergence is very slow. Below is the approximation of  $\pi$  using the first 1,000 terms of the series.

**4 Normal[Series[ArcTan[x], {x, 0, 2000}]] /. {x -> 1.}**

3.14059

## Machin's Formula

John Machin (1680 - 1752) used trigonometric identities to develop a much more efficient formula for approximating  $\pi$ :

$$\frac{\pi}{4} = 4 \arctan\left(\frac{1}{5}\right) - \arctan\left(\frac{1}{239}\right)$$

Machin used this formula in 1706 to compute the first 100 digits of  $\pi$ .

Below is Machin's approximation using the first 10 terms of the Taylor series for the arctan function:

```
f[x_] = Normal[Series[ArcTan[x], {x, 0, 20}]]
```

$$-\frac{x^{19}}{19} + \frac{x^{17}}{17} - \frac{x^{15}}{15} + \frac{x^{13}}{13} - \frac{x^{11}}{11} + \frac{x^9}{9} - \frac{x^7}{7} + \frac{x^5}{5} - \frac{x^3}{3} + x$$

```
N[π, 30]
```

```
4 N[4 f[1/5] - f[1/239], 30]
```

```
3.14159265358979323846264338328
```

```
3.14159265358979169691727961962
```

The difference between the approximation and  $\pi$  is smaller than  $2 \times 10^{-15}$ :

```
Abs[N[π, 30] - 4 N[4 f[1/5] - f[1/239], 30]]
```

```
1.54154536376366 × 10-15
```

#### Problem 1:

How many terms of the series are needed to compute the first 100 digits of  $\pi$ ?

Our estimation of the accuracy above depended on *Mathematica's* ability to compute  $\pi$  accurately to as many digits as desired. Alternatively, one can use an error estimate for the Taylor polynomial.

The series for the second term in Machin's formula will decay much faster than the first one. Thus we can restrict our error estimation to the first term.

The last term of the Taylor expansion above had degree 19. Therefore the error is bounded by the following expression:

$\max \left| \left( \frac{d^{20}(\arctan(y))}{(dy^{20})} \right) \left( \frac{1}{5} \right)^{20} \left( \frac{1}{20!} \right) \right|$ , where the maximum is taken over the interval  $\left[ 0, \frac{1}{5} \right]$ .

```
d[x_] = D[ArcTan[x], {x, 20}]
```


$$\frac{2432902008176640000x}{(x^2+1)^{11}} - \frac{63777066403145711616000x^{19}}{(x^2+1)^{20}} +$$

$$\frac{286996798814155702272000x^{17}}{(x^2+1)^{19}} - \frac{542105064426738548736000x^{15}}{(x^2+1)^{18}} +$$

$$\frac{55804933102752497664000x^{13}}{(x^2+1)^{17}} - \frac{34006131109489803264000x^{11}}{(x^2+1)^{16}} + \frac{124689147401462611968000x^9}{(x^2+1)^{15}} -$$

$$\frac{26719103014599131136000x^7}{(x^2+1)^{14}} + \frac{3082973424761438208000x^5}{(x^2+1)^{13}} - \frac{160571532539658240000x^3}{(x^2+1)^{12}}$$

Plot[Abs[d[y] / 20!], {y, 0, 1 / 5}]



We see that the maximum of the function on the interval  $\left[0, \frac{1}{5}\right]$  is bounded by 0.05. Thus the error for  $\arctan\left(\frac{1}{5}\right)$  is bounded by:

$$\mathbf{err} = 0.05 \left(\frac{1}{5}\right)^{20}$$

$$5.24288 \times 10^{-16}$$

Neglecting the much smaller error stemming from the  $\frac{1}{239}$ -term, we expect our error in approximating  $\pi$  to be bounded by

**4 err**

$$2.09715 \times 10^{-15}$$

**Problem 2:**

Perform an error analysis for the Taylor expansion you are using in Problem 1.

## Derivation of Machin's Formula

The formula is based on the addition formula for the tangent function:

$$\tan(\alpha + \beta) = \frac{\tan(\alpha) + \tan(\beta)}{1 - \tan(\alpha)\tan(\beta)}$$

From this we obtain a formula for  $\tan(4\alpha)$ :

**Expand[Numerator[Together[TrigExpand[Tan[4 α]]]] / Cos[α]^4] /  
Expand[Denominator[Together[TrigExpand[Tan[4 α]]]] / Cos[α]^4]**

$$\frac{4 \tan(\alpha) - 4 \tan^3(\alpha)}{\tan^4(\alpha) - 6 \tan^2(\alpha) + 1}$$

Let's set  $x = \tan(\alpha)$ :

**(Expand[Numerator[Together[TrigExpand[Tan[4 α]]]] / Cos[α]^4] /  
Expand[Denominator[Together[TrigExpand[Tan[4 α]]]] /  
Cos[α]^4]) /. Tan[α] → x**

$$\frac{4x - 4x^3}{x^4 - 6x^2 + 1}$$

If we set  $y = \tan(\beta)$ , and apply the addition formula for the tangent again, we get an expression for  $\tan(4\alpha + \beta)$ :

**Simplify**  $\left[ \left( \frac{4x - 4x^3}{x^4 - 6x^2 + 1} + y \right) / \left( 1 - \frac{4x - 4x^3}{x^4 - 6x^2 + 1} y \right) \right]$

$$\frac{x^4 y - 4x^3 - 6x^2 y + 4x + y}{x^4 + 4x^3 y - 6x^2 - 4xy + 1}$$

We want to use this for  $\frac{\pi}{4} = 4\alpha + \beta$ , with  $\tan\left(\frac{\pi}{4}\right) = 1$ :

**eqn = 1 == Simplify**  $\left[ \left( \frac{4x - 4x^3}{x^4 - 6x^2 + 1} + y \right) / \left( 1 - \frac{4x - 4x^3}{x^4 - 6x^2 + 1} y \right) \right]$

$$1 = \frac{x^4 y - 4x^3 - 6x^2 y + 4x + y}{x^4 + 4x^3 y - 6x^2 - 4xy + 1}$$

We set  $x = \frac{1}{5}$  and solve for  $y$ :

**eqn /. {x → 1/5}**

$$1 = \frac{\frac{476y}{625} + \frac{96}{125}}{\frac{476}{625} - \frac{96y}{125}}$$

**Solve[eqn /. {x → 1/5}, y]**

$$\left\{ \left\{ y \rightarrow -\frac{1}{239} \right\} \right\}$$

Consequently  $\frac{\pi}{4} = 4 \arctan\left(\frac{1}{5}\right) - \arctan\left(\frac{1}{239}\right)$ .

## Other Machin-type Formulas

### Problem 3:

The critical choice Machin made was to set  $x = \frac{1}{5}$ . He then obtained  $y = \frac{1}{239}$  automatically. Why do you think he considered this to be a good choice for  $x$  and  $y$ ?

(It may be helpful to create a numerical table for the first few odd powers of  $\frac{1}{5}$  and  $\frac{1}{239}$ . A hint for why

$y = \frac{1}{239}$  seems to be good choice was already given in the section on Taylor estimates. The next problem may give also you a hint why Machin considered  $x = \frac{1}{5}$  to be the optimal choice.)

### Problem 4:

What happens when you set  $x = \frac{1}{10}$ ? Why does the resulting  $y$  make for a much less suitable "Machin-type" formula?

### Extra-credit Problem 5:

Taylor approximations work best when the approximation happens close to the center of the Taylor series. It therefore seems natural to make both  $x$  and  $y$  as small as possible.

- Find the  $x$  and  $y$  such that the maximum of the two is as small as possible.
- Use this  $x$ - $y$ - combination and a 10-term Taylor expansion to approximate  $\pi$ . Do you get better results than in the top section of this notebook?
- Why did Machin not choose this formula to approximate  $\pi$ ?

## ■ Solution